

# Mesure d'une application (Appscan)

→ exemple : YouTube (en mode non connecté)



```
#Lancement de l'application
measureStart,CHRGT_homePage
applicationStart,com.google.android.youtube
waitUntilText,Home
pause,{PAUSEAFTERLOAD}
measureStop
```

```
#Pause sur la page d'accueil
measureStart,PAUSE_homePage
pause,{PAUSEDURATION}
measureStop
```

```
#Ecriture de la recherche
measureStart,ACTION_homePage_writeResearch
clickByPercent,80,5
enterText,greenspector
pause,{PAUSEAFTERLOAD}
measureStop
```

```
#Recherche d'une vidéo
measureStart,CHRGT_videos
pressEnter
waitUntilText,Home
pause,{PAUSEAFTERLOAD}
measureStop
```

```
#Pause sur la page des vidéos
measureStart,PAUSE_videos
pause,{PAUSEDURATION}
measureStop
```

```
#Scroll sur la page recherchée
measureStart,SCROLL_videos
swipeVertical,90,10,50
swipeVertical,10,90,50
pause,{PAUSEAFTERLOAD}
measureStop
```

```
#Chargement de l'onglet Library
measureStart,CHRGT_libraryTab
clickByTextExact,Library
waitUntilTextExact,Enjoy your favourite videos
pause,{PAUSEAFTERLOAD}
measureStop
```

```
#Pause sur la page de l'onglet Library
measureStart,PAUSE_libraryTab
pause,{PAUSEDURATION}
measureStop
```

```
#Arrêt de l'application
applicationKill,com.google.android.youtube
```

## \*Bonne pratique n°1 : Le nom des étapes

- CHRGT\_[nomVue] : chargement
- ACTION\_[nomVue]\_[nomAction] : remplissage d'un champ
- SCROLL\_[nomVue] : défilement d'une page
- PAUSE\_[nomVue] : attente sur une page

## \*Bonne pratique n°2 : Mesures de pause

Ajouter une pause sur les pages où l'on souhaite mesurer l'état inactif à l'aide de la macro PAUSEDURATION (variable à placer à 30 secondes en extras dans le fichier job.yml).

## \*Bonne pratique n°3 : Méthodes click

Eviter l'utilisation des fonctions GDSL clickByXY ou clickByPercent. Cependant, lorsque l'élément sur lequel on souhaite cliquer n'est pas accessible par son texte, sa description ou son identifiant, ces fonctions peuvent être utilisées.

## \*Bonne pratique n°4 : Fin de chargement avec waitUntil

Lors du passage dans une nouvelle page, utiliser une fonction GDSL commençant par waitUntil pour attendre un élément de cette nouvelle page. Ce type de fonction peut également être utilisé lorsque de nouveaux éléments s'affichent sur la page.

## \*Bonne pratique n°5 : Défilement d'une page

Pour faire défiler une page, privilégier swipeVertical aux fonctions scrollDownward et scrollUpward car c'est plus précis. Pour gagner encore plus en précision, il est possible d'ajouter un 4ème paramètre avec le temps du swipe (par défaut 500 ms).  
Pour en savoir plus : voir [documentation scroll](#)

## \*Bonne pratique n°6 : Courte pause suite au chargement

Ajouter une pause à chaque mesure de chargement, d'action et de scroll avant « measureStop » à l'aide de la macro PAUSEAFTERLOAD (variable à placer à 1 seconde en extras dans le fichier job.yml).

## \*Bonne pratique n°7 : Élément présent une seule fois

Lorsqu'un élément est présent seulement au premier affichage d'un écran, utiliser une fonction du type waitUntilBeforeClick. Cela permet de cliquer sur un élément seulement s'il est présent. Un booléen donné en paramètre à true permet de ne pas faire échouer le test si l'élément n'est pas présent.

## Spécificités iOS

- Le nom du package est différent sur iOS. Pour trouver le nom des applications installées, taper la commande "ios-deploy -B".
  - La fonction swipeVerticale n'existe pas sur iOS. A la place, il existe les fonctions scrollDown, scrollUp et swipe.
- Voir [documentation Android](#) / Voir [documentation iOS](#)

# Mesure d'un site web (Webscan) → exemple : Nantes Métropole



# Préparation de l'environnement de mesure

```
setBrowser,chrome  
applicationKill,com.android.chrome  
browserPrepareAndOpenForReference
```

# Mesure de l'inactivité de référence d'une tab vide de Chrome

```
measureStart,PAUSE_referenceBrowser  
pause,{PAUSEDURATION}  
measureStop
```

# Mise en place de l'adresse dans la barre de navigation

```
browserGoToUrl,https://metropole.nantes.fr
```

#Lancement du site web

```
measureStart,CHRGT_homePage  
pressEnter  
waitUntilText,accept all  
pause,{PAUSEAFTERLOAD}  
measureStop
```

#Acceptation des cookies

```
measureStart,CHRGT_homePageAfterCookies  
clickByText,accept all  
waitUntilText,Actu  
pause,{PAUSEAFTERLOAD}  
measureStop
```

#Pause sur la page d'accueil

```
measureStart,PAUSE_homePage  
pause,{PAUSEDURATION}  
measureStop
```

#Lecture de la page d'accueil

```
measureStart,SCROLL_homePage  
swipeVertical,90,10,50  
swipeVertical,10,90,50  
pause,{PAUSEAFTERLOAD}  
measureStop
```

#Chargement de l'onglet Actu

```
measureStart,CHRGT_actuPage  
clickByText,Actu  
waitUntilText,Toutes les actualités  
pause,{PAUSEAFTERLOAD}  
measureStop
```

#Pause sur la page Actu

```
measureStart,PAUSE_actuPage  
pause,{PAUSEDURATION}  
measureStop
```

## \*Bonne pratique n°8 : Préparation d'une mesure web

Lors d'une mesure d'un site web, préparer l'environnement de mesure avec la fonction `browserPrepareAndOpenForReference`. Cela permet de nettoyer le cache et les onglets. Cela réalise une pause de une minute sur la page « `about:blank` » du navigateur.

## \*Bonne pratique n°9 : Eviter `waitUntilPageLoaded`

Lors d'une mesure d'un site web, éviter l'utilisation de la fonction `waitUntilPageLoaded`. En effet, toutes les données ne sont pas forcément chargées après cette fonction. Cela dépend de la qualité du réseau. Il est préférable d'utiliser une fonction comme `waitUntilText` ou `waitUntilId`.

## \*Bonne pratique n°10 : Gestion des cookies

Accepter les cookies des sites web dans une mesure à part.

## \*Bonne pratique n°11 : Description accessible

La fonction `clickByText` effectue une recherche par texte, puis par description. Elle est donc utile quand la description est accessible.

## \*Bonne pratique n°12 : Téléphone en anglais

En cas de mesure en local, passer le téléphone en anglais car les téléphones du testbench sont en anglais.

## \*Bonne pratique n°13 : Formulaire - Désactiver le clavier

Lors du remplissage des formulaires, la fonction `pressBack` permet de désactiver le clavier. Cela permet aussi de sortir du mode plein écran des vidéos.

## \*Bonne pratique n°14 : Formulaire - Sélection d'un champ

Lors du remplissage des formulaires, si les champs ne sont pas accessibles par leur texte, leur description ou leur identifiant, il est possible d'utiliser la fonction `clickOnFirstEmptyEditText`. Cela permet de sélectionner le prochain champ de texte vide.

## \*Limites de la fonction `enterText` :

Après la fonction `enterText`, ne pas écrire de caractères spéciaux ni accentués car ils ne sont pas pris en compte. Cette fonction peut aussi inverser des numéros lors du remplissage d'un numéro de téléphone dans un champ. Pour contourner ce problème, il est possible d'utiliser les fonctions commençant par `setText`.

## Spécificités iOS

- Les fonctions de gestion du navigateur fonctionnent seulement avec certaines versions de Google Chrome sur iOS. Si cela ne fonctionne pas, il faut réaliser les actions de nettoyage du cache et des onglets à la main avant de lancer les mesures.
- La fonction `clickByText` effectue une recherche par label, puis par name.